
Pendahuluan

Melalui dokumen ini, kalian akan:

• Memahami penggunaan perulangan yang bersarang.

• Memecahkan beberapa persoalan dengan perulangan.

Motivasi: Pola 0

• Pak Dengklek sedang mengajari bebek-bebeknya yang baru
lahir dari telur untuk menggambar.

• Pak Dengklek akan memberikan sebuah bilangan, misalnya N.

• Para bebek diminta untuk mencetak karakter bintang (*)
yang tersusun N baris.

• Contoh untuk N = 3:

*

*

*

Motivasi (lanj.)

• Tentu saja solusinya sederhana, cukup gunakan salah satu
struktur perulangan yang kalian kuasai.

• Untuk kasus ini, lebih tepat digunakan for:

for i := 1 to N do begin

writeln(’*’);

end;

Motivasi: Pola 1

• Setelah para bebek mampu mengerjakan ”Pola 0”, Pak
Dengklek memberikan persoalan yang sedikit lebih sulit.

• Diberikan dua bilangan, misalnya N dan M.

• Cetak karakter bintang (*) yang tersusun N baris dan M
kolom!

• Contoh untuk N = 3 dan M = 5:

*****

*****

*****

• Kali ini, untuk setiap barisnya kita perlu melakukan
perulangan untuk mencetak M karakter bintang!

Contoh Program: pola1 1.pas
• Kita bisa membuat ”for di dalam for”, sehingga membentuk
struktur yang bersarang.

var

N, M: longint;

i, j: longint;

begin

readln(N, M);

for i := 1 to N do begin

for j := 1 to M do begin

write(’*’);

end;

writeln;

end;

end.

Contoh Program: pola1 2.pas
• Tentu saja kita bisa melakukannya dengan struktur
perulangan yang lain, misalnya while:

var

N, M: longint;

i, j: longint;

begin

readln(N, M);

for i := 1 to N do begin

j := 1;

while (j <= M) do begin

write(’*’);

j := j + 1;

end;

writeln;

end;

end.

Contoh Lain: Pola 2

• Soal ”Pola 1” dapat diselesaikan bebek-bebek muda dengan
mudah. Dengan demikian Pak Dengklek memberikan soal
yang lebih menantang.

• Diberikan sebuah bilangan, misalnya N.

• Cetak ”struktur segitiga rata kiri” yang terdiri dari N baris.

• Misalnya untuk N = 5, hasilnya adalah:

*

**

***

****

*****

Contoh Solusi: pola2.pas
• Berikut ini adalah contoh solusinya, dimodifikasi dari
pola1 1.pas:

var

N: longint;

i, j: longint;

begin

readln(N);

for i := 1 to N do begin

for j := 1 to i do begin

write(’*’);

end;

writeln;

end;

end.

Latihan: Pola 3

• Pak Dengklek kemudian memberikan tugas bagi para bebek
muda untuk dikerjakan di rumah nanti.

• Diberikan sebuah bilangan, misalnya N.

• Cetak ”struktur segitiga rata kanan” yang terdiri dari N baris.

• Misalnya untuk N = 5, hasilnya adalah:

____*

___**

__***

_****

*****

• Petunjuk: cetak bagian kiri terlebih dahulu!

Break & Continue

• Kadang kala, kita membutuhkan suatu perulangan untuk
diberhentikan secara paksa atau lompat ke iterasi berikutnya
pada for.

• Pascal menyediakan kedua fitur tersebut, yaitu dengan kata
kunci break dan continue.

Break & Continue (lanj.)

Break

Penggunaan break akan membuat program keluar dari perulangan
yang mengandung kata kunci tersebut.

Continue

Penggunaan continue akan membuat program kembali ke baris
paling atas perulangan, yaitu baris ”for .. to .. do” (untuk for),
”while ... do” (untuk while), dan ”repeat ... until”(untuk repeat).

Contoh Soal: Berhitung 1
• Setelah para bebek mahir dalam menggambar pola, kini Pak
Dengklek ingin mengajarkan mereka tentang berhitung.

• Pak Dengklek akan memberikan dua bilangan, yaitu N dan M.
• Para bebek diminta untuk menuliskan bilangan dari 1 sampai
dengan N. Namun, ketika bilangan yang hendak ditulis adalah
M, jangan cetak bilangan itu dan jangan cetak bilangan
apapun lagi.

• Setelah selesai mencetak bilangan, cetak ”selesai”.
• Contoh untuk N = 10 dan M = 5:

1

2

3

4

selesai

Contoh Program: break.pas
• Berikut ini adalah contoh solusi dari soal ”Berhitung 1”.

var

N, M: longint;

i: longint;

begin

readln(N, M);

for i := 1 to N do begin

if (i = M) then begin

break;

end;

writeln(i);

end;

writeln(’selesai’);

end.

14/31



Penjelasan Program: break.pas

• Ketika break ditemui, perulangan ”for ... to ... do” akan
diberhentikan secara paksa dan lanjut mengeksekusi perintah
selanjutnya, yaitu mencetak tulisan ”selesai”.

Contoh Soal: Berhitung 2

• Kali ini Pak Dengklek mengubah soalnya: diberikan dua
bilangan, yaitu N dan M.

• Para bebek diminta untuk menuliskan bilangan dari 1 sampai
dengan N. Namun, ketika bilangan yang hendak ditulis adalah
kelipatan dari M, jangan cetak bilangan itu.

• Setelah selesai mencetak bilangan, cetak ”selesai”.

• Contoh untuk N = 10 dan M = 2:

1

3

5

7

9

selesai

16/31



Contoh Program: continue.pas
• Berikut ini adalah contoh solusi dari soal ”Berhitung 2”.

var

N, M: longint;

i: longint;

begin

readln(N, M);

for i := 1 to N do begin

if (i mod M = 0) then begin

continue;

end;

writeln(i);

end;

writeln(’selesai’);

end.

Penjelasan Program: continue.pas

• Ketika continue ditemui, eksekusi perintah di dalam ”for ... to
... do” untuk pencacah tersebut langsung dilewati dan lanjut
ke pencacah selanjutnya.

• Artinya, untuk N = 10 dan M = 2, ketika nilai i = 2 dan
continue ditemui, eksekusi akan dilewati dan lanjut ke
perintah paling awal dalam perulangan dengan i = 3.

Contoh Soal: Tes Keprimaan

• Kemudian Pak Dengklek beralih persoalan yang lain, yaitu
tentang teori bilangan.

• Diberikan sebuah bilangan positif yang lebih dari 1, misalnya
N.

• Suatu bilangan N dikatakan prima apabila N positif dan
hanya habis dibagi oleh 1 dan dirinya sendiri.

• Jika N prima, cetak ”<N> adalah bilangan prima” dan jika
tidak, cetak ”<N> bukan bilangan prima”.

Bagaimana kalian akan menyelesaikan persoalan ini?

Solusi 1

• Salah satu solusi yang sederhana adalah: periksa semua
bilangan di antara 2 sampai dengan N-1.

• Jika ada setidaknya satu bilangan yang habis membagi N,
artinya N bukan prima.

Solusi 1: prima1 1.pas
var

N, i: longint;

prima: boolean;

begin

readln(N);

prima := TRUE;

for i := 2 to N-1 do begin

if (N mod i = 0) then begin

prima := FALSE;

end;

end;

if (prima) then begin

writeln(N, ’ adalah bilangan prima’);

end else begin

writeln(N, ’ bukan bilangan prima’);

end;

Solusi 2

• Solusi 1 melakukan pemeriksaan dari 2 sampai dengan N-1,
artinya dibutuhkan pemeriksaan sebanyak N-2 kali.

• Sebetulnya pemeriksaan bisa dihentikan ketika ditemukan
setidaknya satu saja bilangan yang habis membagi N.

• Dengan demikian bisa digunakan break untuk
memberhentikan perulangan begitu ditemukan bilangan yang
habis membagi N.

Solusi 2: prima1 2.pas
var

N, i: longint;

prima: boolean;

begin

readln(N);

prima := TRUE;

for i := 2 to N-1 do begin

if (N mod i = 0) then begin

prima := FALSE;

break;

end;

end;

if (prima) then begin

writeln(N, ’ adalah bilangan prima’);

end else begin

writeln(N, ’ bukan bilangan prima’);

end;

end.

Penggunaan Break

• Dalam prakteknya, penggunaan break kurang dianjurkan
karena membuat program menjadi ”kotor”.

• Jika membutuhkan perulangan yang suatu ketika perlu
diberhentikan (sudah memenuhi syarat tertentu), gunakan
while atau repeat.

• Solusi yang lebih baik ditunjukkan pada ”prima1 3.pas”.

Solusi 3: prima1 3.pas

var

N, i: longint;

found: boolean;

begin

readln(N);

found := FALSE;

i := 2;

while ((i < N) and not found) do begin

if (N mod i = 0) then begin

found := TRUE;

end;

i := i + 1;

end;

Solusi 3: prima1 3.pas (lanj.)

if (not found) then begin

writeln(N, ’ adalah bilangan prima’);

end else begin

writeln(N, ’ bukan bilangan prima’);

end;

end.

Contoh Soal: Pembangkit Prima

• Kini para bebek sudah mengetahui bagaimana memeriksa
keprimaan suatu bilangan. Pak Dengklek memutuskan untuk
memberikan tambahan soal untuk pekerjaan rumah.

• Diberikan sebuah bilangan bulat N. Pak Dengklek meminta
bebek-bebeknya untuk menuliskan N bilangan prima pertama.

• Contoh untuk N = 5:

2

3

5

7

11

Solusi: Pembangkit Prima

• Salah satu strategi yang dapat kalian gunakan adalah ”selama
belum ditemukan N bilangan prima, cari bilangan prima!”.

• Bagaimana mencari bilangan prima? Coba saja dari 2, 3, 4,
dan seterusnya sampai ditemukan N bilangan prima.

Contoh Solusi: prima2.pas
var

N, cur, i: longint;

count: longint;

found: boolean;

begin

readln(N);

count := 0; (* banyaknya prima yang sudah ditemukan *)

cur := 2; (* nilai yang akan diperiksa keprimaannya *)

while (count < N) do begin

found := FALSE;

i := 2;

while ((i < cur) and not found) do begin

if (cur mod i = 0) then begin

found := TRUE;

end;

i := i + 1;

end;

Contoh Solusi: prima2.pas (lanj.)

if (not found) then begin

(* ditemukan prima!

cetak dan tambahkan prima yg sudah ditemukan *)

writeln(cur);

count := count + 1;

end;

(* entah ini prima atau bukan, lanjut untuk

memeriksa bilangan berikutnya *)

cur := cur + 1;

end;

(* keluar dari while, dipastikan count = N *)

end.

Penutup

• Percabangan dan perulangan merupakan dua struktur kontrol
yang sangat penting pada pemrograman.

• Kalian diharapkan berlatih sampai lancar di kedua hal
tersebut, baru lanjut untuk mempelajari materi selanjutnya.

